

Laboratorium Metod Numerycznych

Równania różniczkowe

Laboratorium 7

Opis zjawisk fizycznych charakteryzujących się zmianą w czasie jest realizowany za pomocą równań różniczkowych. Przykładem takich zjawisk w elektrotechnice są stany nieustalone w obwodach elektrycznych występujące w wyniku skokowej zmiany parametrów układu. Analityczne rozwiązanie problemu jest często procesem bardzo złożonym i czasochłonnym. Rozwiązaniem może być zastosowanie jednej z opracowanych metod numerycznych do wyznaczenia przybliżonych wartości rozwiązania równania różniczkowego. W odróżnieniu od omawianych wcześniej metod dedykowanych rozwiązywaniu układów równań liniowych i nieliniowych gdzie wyznaczano przybliżona wartość rozwiązania równań, w przypadku równań różniczkowych odpowiedzią jest tablica wartości opisujących przebieg funkcji będącej rozwiązaniem równania różniczkowego.

W ramach ćwiczenia omówione zostaną trzy metody wyznaczania rozwiązania równania różniczkowego:


- wzór *Taylor'a*,
- metoda *Euler'a*,
- metody *Runge-Kutty'ego*,

i układów równań różniczkowych, I i II rzędu. W środowiskach obliczeniowych zazwyczaj zaimplementowane są funkcje realizujące rozwiązywanie równań różniczkowych. Metody te wyznaczają kolejne wartości funkcji stanowiącej rozwiązanie równania różniczkowego w zadanym przedziale obliczeń. Kolejne kroki obliczane są na podstawie wcześniej wyznaczonych wartości. Warunkiem koniecznym jest więc podanie wartości początkowych, stanowiących warunki brzegowe zadania. W przypadku stanów nieustalonych w obwodach elektrycznych będą to np. wartości napięcia na kondensatorze, prądu w gałęzi z cewką czy czasu definiującego chwile zmiany stanu obwodu. Istotnym czynnikiem wpływającym na dokładność uzyskanych wyników będzie krok obliczeń, np. Δt . Im mniejsza to będzie wartość, tym dokładniejsze wyniki zostaną przez metodę zwrócone.

W środowisku **Scilab** jest to funkcja `ode()`, której zastosowanie także zostanie omówione. Możliwe jest także zastosowanie narzędzi programowania graficznego **XCOS**, gdzie za pomocą algorytmu opracowanego na podstawie równania różniczkowego za-

pisanego za pomocą zmiennych stanu można uzyskać jego rozwiązanie. Ten sposób programowania obliczeń numerycznych także zostanie w instrukcji omówiony.

Metody numerycznego rozwiązywania równań różniczkowych zostaną przetestowane na przykładzie równania różniczkowego w granicach opisującego zmianę napięcia na kondensatorze w szeregowym obwodzie RC załączanym na źródło napięcia stałego E , jak pokazano na Rys.1.



Rys. 1: Testowy obwód RC

Równanie różniczkowe dla analizowanego obwodu można wyprowadzić wychodząc z praw Kirchhoffa, jak pokazano poniżej.

$$E = u_R + u_C \quad (1)$$

$$i = C \frac{du_c}{dt} \rightarrow E = RC \frac{du_c}{dt} + u_C \quad (2)$$

$$\frac{du_c}{dt} = \frac{E - u_c}{RC} = \frac{E}{RC} - \frac{u_c}{RC} \quad (3)$$

Dla przykładowego zadania przyjęto parametry obwodu z Rys.1 $R = 2000\Omega$, $E = 10V$; $C = 25\mu F$. Analitycznie wyznaczone rozwiązanie równania różniczkowego (3) przyjmuje postać zapisaną poniższym wzorem.

$$u_C = E - E \cdot e^{-\frac{t}{RC}} \quad (4)$$

1) Wzór Taylor'a

Podstawowym sposobem wyznaczenia przybliżonego rozwiązania równania różniczkowego jest zastosowanie wzoru Taylor'a. Zastosowanie metody wymaga aby możliwe było zapisanie pochodnych cząstkowych wyższych rzędów równania różniczkowego. Im wyższy stopień możliwy do wyznaczenia tym dokładniejsze wyniki zwracane będą przez metodę, przy tej samej wartości korku obliczeniowego Δx . Obliczenia realizowane są w stałym kroku iteracyjnym zdefiniowanym przez wartość Δx (zazwyczaj, dla obliczeń realizowanych w przestrzeni czasu, będzie to Δt). Ogólna postać wzoru Taylor'a ma postać zapisaną poniżej.

$$y(x + \Delta x) = y(x) + \sum_{n=1}^m \frac{1}{n!} \cdot (\Delta x)^n \cdot y^{(n)} \quad (5)$$

gdzie: m jest rzędem przybliżenia, a $y^{(n)}$ jest pochodną cząstkową rzędu n .


Wybór rzędu przybliżenia jest często zależny od możliwych do zapisania rzędów pochodnych cząstkowych równania różniczkowego. Dla przykładowego równania (3), możliwe jest tylko zapisanie pochodnej cząstkowej II rzędu, jak pokazano poniżej.

$$\frac{d^2 u_c}{dt^2} = -\frac{1}{RC} \quad (6)$$


W oparciu o algorytm (5) opracować można skrypt wyznaczający przebieg funkcji będącej rozwiązaniem równania różniczkowego (3), jak pokazano poniżej.

```
clear;clc;clf;
R=2e3; E=10; C=25e-6; tau=R*C;
n=25; t0=0; tn=5*tau; dt=(tn-t0)/n;
t=[t0]; u=[0]; i=[0];
function u=rozwiazanie(t)
    u=E*(1-exp(-t/tau));
endfunction
for k=1:n-1
    du(1)=E/(R*C)-u(k)/(R*C);
    du(2)=-1/(R*C)
    u(k+1)=u(k);
    for j=1:2
        u(k+1)=u(k+1)+(1/factorial(j))*(dt^j)*du(j);
    end
    t(k+1)=t(k)+dt;
    i(k+1)=C*du(1)*1000;
    plot2d(t,u,5);
    plot2d(t,rozwiazanie(t),-4);
end
xlabel('czas t [s]')
ylabel('Napięcie na kondensatorze uc [V]')
os1=gca(); os2=newaxes();
plot2d(t,i,6);
os2.filled="off"; os2.axes_visible(1)="off";
os2.y_location="right";
ylabel('Prąd w obwodzie i [mA]')
```

Na Rys.2 pokazano przebiegi uzyskane z wzoru Taylor'a uzyskane dla rozwinięć pierwszego i drugiego rzędu. Obliczenia zrealizowano dla 25 kroków ($\Delta t = 0.01$). Krzywą czerwoną zaznaczono przebieg uzyskany z rozwiązania analitycznego. Zmniejszenie kroku obliczeń w sposób znaczący wpływa na dokładność obliczeń. Na Rys.3 pokazano zestawienie rozwiązań po zwiększeniu liczby kroków, a więc zmniejszeniu długości kroku, do 250 kroków.



Rys. 2: Rozwiązanie równania różniczkowego (3) wzorem Taylor'a



Rys. 3: Zestawienie przebiegów dla 25 i 100 kroków obliczeń

Zauważyć można, że zwiększenie liczby kroków symulacji w sposób znaczący poprawiło wyniki obliczeń, przybliżając je do wyników analitycznych. Zastosowanie metody może być kłopotliwe ze względu na konieczność wyznaczania pochodnych cząstkowych równania różniczkowego. Co może być bardzo złożonym zagadnieniem w zależności od budowy równania różniczkowego. W innych sytuacjach wyższe rozwinięcia równania różniczkowego mogą być równe 0, co będzie wpływało na dokładność uzyskiwanych wyników, wymuszając znaczne zmniejszenie kroku obliczeń, co w konsekwencji wpływa na wydłużenie obliczeń.

2) Metoda Euler'a

Jest najprostszą z metod opartych na wzorze Taylor'a, w której ograniczono się wyłącznie do równania różniczkowego, a pomija równania cząstkowe. Jest ona dedykowana do rozwiązywania równań różniczkowych I rzędu. Dokładność uzyskiwanych z niej wyników jest bezpośrednio zależna od długości kroku obliczeniowego, im krótszy krok symulacji tym dokładniejsze wyniki obliczeń. Wyznaczanie kolejnych punktów przebiegu rozwiązania równania różniczkowego realizowana jest zgodnie z zapisanym poniżej równaniem.

$$y(x + \Delta x) = y(x) + \Delta x \cdot \frac{dy(x)}{dx} \quad (7)$$

Implementacja metody Euler'a w kodzie języka programowania wymaga niewielkiej modyfikacji kodu zapisanego na podstawie równań Taylor'a, jak pokazano w przykładowym kodzie zapisanym poniżej.


```
clear;clc;clf;
R=2e3;E=10;C=25e-6; tau=R*C;
n=25; t0=0; tn=5*tau; dt=(tn-t0)/n;
t=[t0]; u=[0]; i=[0];
```

```

function u=rozwiazanie(t)
    u=E*(1-exp(-t/tau));
endfunction
for k=1:n-1
    du=E/(R*C)-u(k)/(R*C);
    u(k+1)=u(k)+dt*du;
    t(k+1)=t(k)+dt;
    i(k+1)=C*du*1000;
end
plot2d(t,u,5);
plot2d(t,rozwiazanie(t),-4);
xlabel('czas t [s]')
ylabel('Napięcie na kondensatorze uc [V]')
os1=gca(); os2=newaxes();
plot2d(t,i,6);
os2.filled="off"; os2.axes_visible(1)="off";
os2.y_location="right";
ylabel('Prąd w obwodzie i [mA]')

```

Wystarczy skasować odwołania do wyższych pochodnych cząstkowych równania różniczkowego. Dla przykładowego równania różniczkowego (3) z metody Euler'a uzyskano rozwiązanie pokazane na Rys.4 dla korków obliczeniowych $\Delta t = 0.01s$ i $\Delta t = 0.001s$.



Rys. 4: Rozwiązanie równania różniczkowego z metody Euler'a

Można zauważyć że dziesięciokrotne zmniejszenie długości kroku znacząco poprawiło dokładność uzyskanych wyników obliczeń, przebieg obliczeniowy pokrywa się z przebiegiem wzorcowym (4).

3) Metody Runge-Kutty'ego

Podstawowa idea metod polega na przybliżeniu rozwiązania równań różniczkowych, dzieląc krok czasowy na mniejsze kroki i obliczając wartości funkcji na każdym z tych kroków. Wraz ze wzrostem stopnia metody rośnie gęstość podziału kroku czasowego.

Uogólniona procedura rozwiązywania równania różniczkowego metodą Runge-Kutty'ego realizowana jest zgodnie z zapisanym poniżej algorytmem.

1. Obliczenie wartości pochodnych funkcji na początku kroku czasowego.
2. Obliczenie wartości pochodnych w połowie kroku czasowego.
3. Obliczenie wartości pochodnych w połowie kroku czasowego, ale korzystając z pochodnych obliczonych w poprzednich krokach.
4. Obliczenie wartości pochodnych na końcu kroku czasowego, korzystając z pochodnych z poprzednich kroków.
5. Aktualizacja wartości funkcji na podstawie obliczonych pochodnych.

Za pomocą zapisu matematycznego powyższy algorytm można zapisać:

$$y_{n+1} = y_n + \sum_{i=1}^m w_i \cdot k_i \quad (8)$$

$$\begin{cases} \text{dla } i = 1 \longrightarrow k_i = & \Delta x \cdot f(x_n, y_n) \\ \text{dla } i > 1 \longrightarrow k_i = & \Delta x \cdot f\left(x_n + a_i \Delta x, y_n + \sum_{j=1}^{i-1} b_{ij} \cdot k_j\right) \end{cases} \quad (9)$$

gdzie: m jest rzędem metody (1,2,3, i 4), w , a i b stałymi zależnymi od rzędu metody. W tabeli 1 zestawiono wartości stałej w i współczynników k_i w zależności od rzędu metody.

Tab. 1: Zestawienie współczynników metody Runge-Kutty'ego

Rząd	w_i	k_i	Nazwa
I	$w_1 = 1$	$k_1 = \Delta x \cdot f(x_n, y_n)$	Euler'a
II	$w_1 = \frac{1}{2}$ $w_2 = \frac{1}{2}$	$k_1 = \Delta x \cdot f(x_n, y_n)$ $k_2 = \Delta x \cdot f(x_n + \Delta x, y_n + k_1)$	Heun'a
III	$w_1 = \frac{1}{6}$ $w_2 = \frac{2}{3}$ $w_3 = \frac{1}{6}$	$k_1 = \Delta x \cdot f(x_n, y_n)$ $k_2 = \Delta x \cdot f(x_n + \frac{1}{2}\Delta x, y_n + \frac{1}{2}k_1)$ $k_3 = \Delta x \cdot f(x_n + \Delta x, y_n - k_1 + 2k_2)$	Simpson'a
IV	$w_1 = \frac{1}{6}$ $w_2 = \frac{1}{3}$ $w_3 = \frac{1}{3}$ $w_4 = \frac{1}{6}$	$k_1 = \Delta x \cdot f(x_n, y_n)$ $k_2 = \Delta x \cdot f(x_n + \frac{1}{2}\Delta x, y_n + \frac{1}{2}k_1)$ $k_3 = \Delta x \cdot f(x_n + \frac{1}{2}\Delta x, y_n + \frac{1}{2}k_2)$ $k_4 = \Delta x \cdot f(x_n + \Delta x, y_n + k_3)$	Runge-Kutty'ego


Ze względu na najwyższą dokładność obliczeniową zazwyczaj stosowana w obliczeniach numerycznych równań różniczkowych jest metoda Runge-Kutty'ego rzędu czwartego, jak pokazano poniżej w zestawieniu wyników zrealizowanych dla równania

przykładowego (3). Kod przykładowego skryptu obliczeniowego realizującego rozwiązanie równania różniczkowego (3) pokazano poniżej. Metodę Runge-Kutty'ego rzędu czwartego zaimplementowano w funkcji. W sposób analogiczny można zaprogramować metody niższych rzędów.


```
clear;clc;clf;
R=2e3;E=10;C=25e-6; tau=R*C;
n=25; t0=0; tn=5*tau; dt=(tn-t0)/n;
t=[t0]; u=[0]; i=[0];
function u=rozwiazanie(t)
    u=E*(1-exp(-t/tau));
endfunction
function y=rownanie(x)
    y=E/(R*C)-x/(R*C);
endfunction
function [x,y]=RK4(dx,xw,yw)
    k1=dx*rownanie(yw);
    k2=dx*rownanie(yw+0.5*k1);
    k3=dx*rownanie(yw+0.5*k2);
    k4=dx*rownanie(yw+k3)
    y=yw+(1/6)*k1+(1/3)*k2+(1/3)*k3+(1/6)*k4;
    x=xw+dx;
endfunction
for k=1:n
    [t(k+1),u(k+1)]=RK4(dt,t(k),u(k));
    i(k+1)=C*rownanie(u(k))*1000;
end
plot2d(t,u,5);
plot2d(t,rozwiazanie(t),-4);
xlabel('czas t [s]')
ylabel('Napięcie na kondensatorze uc [V]')
os1=gca();
os2=newaxes();
plot2d(t,i,6);
os2.filled="off"; os2.axes_visible(1)="off";
os2.y_location="right";
ylabel('Prąd w obwodzie i [mA]')
legends(["uC" "i" "uC analityczne"],[5 6 -4],opt="lr");
```

Jak można zauważyć na Rys.5 największy błąd jest osiągnięty dla rzędu pierwszego. W przypadku pozostałych metod różnica jest trudno zauważalna, jednakże po zbliżeniu przebiegu na wykresie w pobliżu wybranego punktu obliczeniowego można zauważyć istotne różnice w przebiegach rozwiązań dla poszczególnych stopni metody, jak pokazano na Rys.6. Dziesięciokrotne zmniejszenie kroku obliczeniowego sprawia


że różnice pomiędzy przebiegiem rozwiązania analitycznego a otrzymanymi z obliczeń numerycznych metodami Runge-Kutty'ego praktycznie się zacierają, szczególnie dla wyższych rzędów, co pokazano na Rys.7 i Rys.8.




Rys. 5: Rozwiązanie równania różniczkowego metodą Runge-Kutty'ego rzędu 1-4 dla $\Delta t = 0.01$



Rys. 6: Różnice pomiędzy rzędem 2-4 a przebiegiem analitycznym



Rys. 7: Rozwiązanie równania różniczkowego metodą Runge-Kutty'ego rzędu 1-4 dla $\Delta t = 0.001$




Rys. 8: Różnice pomiędzy rzędem 3-4 a przebiegiem analitycznym

4) Układ równań różniczkowych i równania wyższych rzędów

Opisane powyżej metody dedykowane są dla równań różniczkowych I rzędu. Obliczenia mogą być realizowane także dla układu równań różniczkowych I rzędu. W przypadku równań lub układów równań wyższych rzędów należy je sprowadzić do układu równań I rzędu.

Ponieważ metody jedno-krokowe wyznaczają kolejne wartości na podstawie wcześniej wyznaczonych, zapisanie modelu numerycznego rozwiązującego układ równań

różniczkowych I rzędu wiąże się z zapisaniem ich w modelu po kolei. Zastosowanie poszczególnych metod rozwiązywania układów równań różniczkowych zostanie pokazana na przykładzie układu równań II rzędu zapisanych dla opisanego stanu nieustalonego w szeregowym obwodzie RLC załączanym na źródło napięcia stałego, jak pokazano na Rys.9.



Rys. 9: Testowy obwód RLC

Parametry testowego obwodu przyjmują wartości: $E=100\text{ V}$, $R=0.25\ \Omega$, $L=25\text{mH}$ i $C=33\text{mF}$. Układ pokazany na Rys.9 można opisać układem równań różniczkowych wychodząc z równania opisującego dany układ zgodnie z II prawem Kirchhoffa, jak pokazano poniżej,

$$E = R \cdot i + u_L + u_C \quad (10)$$

Ponieważ prąd w obwodzie z kondensatorem jest opisany zależnością

$$i = C \cdot \frac{du_C}{dt} \quad (11)$$

a spadek napięcia na cewce jest funkcją opisana zależnością

$$u_L = L \cdot \frac{di}{dt} \quad (12)$$

równanie napięciowe obwodu można zapisać

$$E = RC \cdot \frac{du_C}{dt} + L \cdot \frac{di}{dt} + u_C \quad (13)$$

Zapisane równanie różniczkowe dwóch zmiennych można sprowadzić do równania różniczkowego jednej zmiennej wyrażając natężenie prądu w obwodzie funkcją spadku napięcia na kondensatorze.

$$E = RC \cdot \frac{du_C}{dt} + LC \cdot \frac{d^2u_C}{dt^2} + u_C \quad (14)$$

Ze względu na właściwości metod numerycznych stosowanych do rozwiązywania równań różniczkowych konieczne jest sprowadzenie równania różniczkowego II rzędu

(14) do układu równań I rzędu dwóch niewiadomych. Łącząc i przekształcając równania (10,11 i 12) można zapisać równoważny układ równań, jak pokazano poniżej.

$$\begin{cases} \frac{di}{dt} = \frac{E - Ri - u_C}{L} \\ \frac{du_C}{dt} = \frac{i}{C} \end{cases} \quad (15)$$


Wzór Taylor'a wymaga zapisania równań różniczkowych wyższych rzędów dla rozwiązywanego równania. Dla przedmiotowego układu równań (15) można zapisać przedstawione poniżej wyższe pochodne.

$$II. \begin{cases} \frac{d^2i}{dt^2} = -\frac{R \frac{di}{dt} + \frac{du_C}{dt}}{L} \\ \frac{d^2u_C}{dt^2} = \frac{1}{C} \cdot \frac{di}{dt} \end{cases} \quad (16)$$


$$III. \begin{cases} \frac{d^3i}{dt^3} = -\frac{R \frac{d^2i}{dt^2} + \frac{d^2u_C}{dt^2}}{L} \\ \frac{d^3u_C}{dt^3} = \frac{1}{C} \cdot \frac{d^2i}{dt^2} \end{cases} \quad (17)$$

$$IV. \begin{cases} \frac{d^4i}{dt^4} = -\frac{R \frac{d^3i}{dt^3} + \frac{d^3u_C}{dt^3}}{L} \\ \frac{d^4u_C}{dt^4} = \frac{1}{C} \cdot \frac{d^3i}{dt^3} \end{cases} \quad (18)$$

Dokładność i zbieżność metody silnie zależy od wartości kroku obliczeniowego i zastosowanego stopnia (wyższych pochodnych) równania. Na Rys.10. pokazano uzyskane przebiegi dla kroku obliczeniowego $\Delta t = 0.01s$ przy zastosowaniu we wzorze tylko I pochodnej (metoda Euler'a) i od I do IV. Przy mniejszej wartości kroku $\Delta t = 0.0001s$ różnice pomiędzy zastosowanym stopniem rozwinięcia są praktycznie niezauważalne, jak pokazano na Rys.11.



Rys. 10: Rozwiązanie równania różniczkowego wzorem Taylora dla rozwinięcia 1 i 4 dla $\Delta t = 0.01s$



Rys. 11: Rozwiązanie równania różniczkowego wzorem Taylora dla $\Delta t = 0.0001s$

Analogiczną metodykę należy zastosować dla metod Runge-Kutty'ego multiplikując liczbę współczynników i równań w zależności od liczby równań różniczkowych w układzie modyfikując pokazany wcześniej skrypt.

5) Funkcja *ode()*

W środowisku obliczeniowym **Scilab** dostępna jest funkcja `ode()` realizująca numeryczne rozwiązanie równania i układów równań różniczkowych. Wyniki zwracane są w postaci tablic wartości przebiegów funkcji stanowiących rozwiązanie równania, tak sama jak w przypadku omówionych wcześniej metod.

Funkcja `ode()`, składnie podstawową zapisano poniżej, wymaga podania czterech atrybutów obowiązkowych aby mogła zwrócić rozwiązanie.

`y=ode(y0,t0,t,funkcja);`

gdzie:

- `y0` → jest wektorem wartości brzegowych zadania (warunki początkowe)
- `t0` → wartość początkowa zmiennej podcałkowej (zazwyczaj czas)
- `t` → wektor wartości poszczególnych kroków czasowych dla jakich wyznaczone będzie rozwiązanie
- `funkcja` → nazwa funkcji w której zdefiniowany jest układ równań różniczkowych w postaci równań stanu

Dla przykładowego układu rozpatrywanego w ćwiczeniu, pokazanym na Rys.9, równania stanu można zapisać na podstawie równań różniczkowych zapisanych poniżej:

$$\begin{cases} \frac{d^2 u_C}{dt^2} = \frac{E - RC \cdot \frac{du_C}{dt} - u_C}{LC} \\ \frac{di}{dt} = \frac{E - Ri - u_C}{L} \\ \frac{du_C}{dt} = \frac{i}{C} \end{cases} \quad (19)$$

przyjmując zmienne stanu w postaci:

$$\begin{cases} x_1 = \frac{d^2 u_C}{dt^2} \\ x_2 = u_C \\ x_3 = i \end{cases} \quad (20)$$

Ostatecznie równania stanu przyjmują postać zapisana poniżej i implementowaną w funkcji skryptu obliczeniowego **Scilab**'a.


$$\begin{cases} y_1 = \frac{E - RC \cdot x_1 - x_2}{LC} \\ y_2 = \frac{E - R \cdot x_3 - x_2}{L} \\ y_3 = \frac{x_3}{C} \end{cases} \quad (21)$$

Kod skryptu obliczającego zdefiniowany równaniami stanu układ równań różniczkowych dla obwodu z Rys.9 pokazano poniżej wraz z uzyskanymi wynikami pokazanymi na Rys.12.

```

clc; clear;clf;
function [y]=R_stanu(t,x)
    y(1)=(E-R*C*x(1)-x(2))/(L*C);
    y(2)=x(3)/C;
    y(3)=(E-R*x(3)-x(2))/L;
endfunction
R=0.25; L=25e-3; C=33e-3; E=100;
dt=0.0001;
koniec=1;
t0=0; y0=[0;0;0];
t=t0:dt:koniec;
y=ode(y0,t0,t,R_stanu);
uc=y(2,:); i=y(3,:);
plot2d(t,[i' uc'],[2 5]);
osie=gca(); osie.x_location="origin"; osie.y_location="origin";
xlabel("Czas t [s]");
legends(["i(t)" "u(t)"],[2 5], opt="ur");

```



Rys. 12: Rozwiązanie równania różniczkowego funkcją ode()

Należy zwrócić uwagę na konieczność dodania atrybutu t w kodzie funkcji równań stanu pomimo że w niej zmienna t nie występuje w sposób jawny. Jej obecność jest wymagana przez funkcję `ode()`.


6) Graficzne programowanie w *XCOS*

Środowisko *Scilab* pozwala także z wykorzystaniem podprogramu *XCOS* na graficzne zaprogramowanie rozwiązań układów równań różniczkowych. W przypadku zadania przykładowego algorytm graficzny można oprzeć na równaniach (19). Programowania graficzne polega na przetworzeniu sygnałów za pomocą dedykowanych bloków z grup:


- Systemy czasu ciągłego (INTEGRAL_m)
- Operacje matematyczne (GAIN_f, BIGSOM_f)
- Trasowanie sygnału (DEMUX_f)
- Źródła (CLOCK_c, CONST_f, GAINSIN_f,...)
- Sinks (CSCOPE lub CMSCOPE)

Przygotowując algorytm graficzny w programie należy zdefiniować warunki zadania (odnoszące się do kroku czasowego i czasu trwania obliczeń) oraz można przeddefiniować stałe (parametry obwodu) aby łatwiejsza była modyfikacja parametrów al-


gorytmu. Konfiguracje tych parametrów realizuje się poprzez pozycje dostępne w zakładce symulacja, jak pokazano na Rys.13.



Rys. 13: Wybór okien konfiguracyjnych projektu




Rys. 14: Ustawienie czasu trwania symulacji




Rys. 15: Definiowanie stałych

Równania różniczkowe opisujące analizowany układ z Rys.9 można w postaci algorytmu graficznego zaprojektować na dwa równoważne sposoby pokazane na Rys.16 i Rys.17.



Rys. 16: Algorytm wersja A



Rys. 17: Algorytm wersja B

Atrybuty poszczególnych elementów schematów można edytować wywołując okno konfiguracyjne wyświetlane po podwójnym kliknięciu elementu. Istotne dla przebiegu symulacji jest właściwe skonfigurowanie zegara i oscyloskopu. W oknie konfiguracyjnym zegara należy ustawić krok symulacji i czas rozpoczęcia symulacji (np.: 0.0001 i 0). W oknie oscyloskopu należy zdefiniować zakresy osi pionowych oraz zakres osi czasowej.

Zadania

Na podstawie niniejszej instrukcji, wiadomości wykładowych i literaturowych rozwiązać układ równań różniczkowych wyprowadzonych dla ustalonego z prowadzącym laboratorium obwodu elektrycznego. W sprawozdaniu należy:

1. zamieścić schemat obwodu wraz z informacjami o jego parametrach
2. wyprowadzenie równań różniczkowych opisujących obwód w stanie nieustalonym
3. zaprojektować funkcje wyznaczające rozwiązanie wyprowadzonych równań różniczkowych:
 - metodą z wykorzystaniem wzoru Taylor'a, o ile to możliwe z wykorzystaniem wyższych pochodnych równań różniczkowych co najmniej do 4 stopnia
 - metoda Runge-Kutty'ego rzędów od 1 do 4
4. zweryfikować uzyskane wyniki za pomocą rozwiązania uzyskanego z funkcji `ode()` lub algorytmu graficznego z podprogramu ***XCOS***
5. Przeanalizować wielkość kroku obliczeniowego i stopnia metody (wzór Taylor'a i metody Runge-Kutty) na dokładność obliczeń
6. Sprawdzić wpływ rezystancji występujących w analizowanym obwodzie na charakter przebiegu (przebieg aperiodyczny lub oscylacyjny), określić wartość rezystancji krytycznej (przebieg aperiodyczny-krytyczny)